```
>
```

Z transform exercise (4 oct 2012)

```
Error, missing operator or `;`
> restart:
```

Example 1 (sum of cubes)

The recurrence

```
> R := S(n+1) = S(n) + (n+1)^3 ;
```

$$R := S(n+1) = S(n) + (n+1)^3 \quad (1.1)$$

The initial condition

```
> INI := S(0) = 0 ;
```

$$INI := S(0) = 0 \quad (1.2)$$

Solve using Maple primitive

```
> SOL := rsolve( {R,INI}, S ) ;
```

$$SOL := -12\,(n+1)\left(\frac{1}{2}\,n+1\right)\left(\frac{1}{3}\,n+1\right)+7\,(n+1)\left(\frac{1}{2}\,n+1\right)+6\,(n+1)\left(\frac{1}{2}\,n+1\right)\left(\frac{1}{3}\,n+1\right)\left(\frac{1}{4}\,n+1\right)-n-1 \quad (1.3)$$

Try to simplify

```
> simplify(SOL) ; factor(%) ;
```

$$\frac{1}{2}\,n^3+\frac{1}{4}\,n^2+\frac{1}{4}\,n^4$$

$$\frac{1}{4}\,n^2\,(n+1)^2 \quad (1.4)$$

Solve the same problem using Z-transform

Z-transform the recurrence

```
> ztrans(R,n,z) ;
```

$$z\,ztrans(S(n),n,z)-S(0)\,z = ztrans(S(n),n,z)+\frac{z\,(z^2+1+4\,z)}{(z-1)^4}+\frac{3\,z\,(z+1)}{(z-1)^3}+\frac{3\,z}{(z-1)^2}+\frac{z}{z-1} \quad (1.5)$$

Decompose (n+1)^3 as a sum of binomial coefficients

```
> simplify(expand((n+1)^3)-n*(n-1)*(n-2)-6*n*(n-1)) ;
```

$$7\,n+1 \quad (1.6)$$

Thus (n+1)^3 = n_3 + 6 n_2 + 7 n + 1

Compute Z-transfor table for binomial coefficients

```
> simplify(ztrans(n*(n-1)*(n-2),n,z)) ;
  simplify(ztrans(n*(n-1),n,z)) ;
  simplify(ztrans(n,n,z)) ;
  simplify(ztrans(1,n,z)) ;
```

$$\frac{6\,z}{(z-1)^4}$$
$$\frac{2\,z}{(z-1)^3}$$
$$\frac{z}{(z-1)^2}$$
$$\frac{z}{z-1} \tag{1.7}$$

The transforrmed problem is

```
> Rz := ztrans(S(n+1),n,z) = ztrans(S(n),n,z) +
                             simplify(ztrans(n*(n-1)*(n-2),n,z)) +
                             6*simplify(ztrans(n*(n-1),n,z)) +
                             7*simplify(ztrans(n,n,z)) +
                             simplify(ztrans(1,n,z)) ;
```

$$Rz := z\,ztrans(S(n), n, z) - S(0)\,z = ztrans(S(n), n, z) + \frac{6\,z}{(z-1)^4} + \frac{12\,z}{(z-1)^3} + \frac{7\,z}{(z-1)^2} + \frac{z}{z-1} \tag{1.8}$$

Apply Boundary conditions

```
> Rz0 := subs( INI, Rz ) ;
```

$$Rz0 := z\,ztrans(S(n), n, z) = ztrans(S(n), n, z) + \frac{6\,z}{(z-1)^4} + \frac{12\,z}{(z-1)^3} + \frac{7\,z}{(z-1)^2} + \frac{z}{z-1} \tag{1.9}$$

Solve the recurrence in the domain z (Maple destroy the already done partial fraction expansion)

```
> RzSolved := solve( Rz0, ztrans(S(n),n,z) ) ;
```

$$RzSolved := \frac{z^2\left(z^2+1+4\,z\right)}{(z-1)^5} \tag{1.10}$$

Redo partial fraction expansion

```
> expand(convert( RzSolved/z, parfrac )*z) ;
```

$$\frac{7\,z}{(z-1)^3} + \frac{z}{(z-1)^2} + \frac{6\,z}{(z-1)^5} + \frac{12\,z}{(z-1)^4} \tag{1.11}$$

Invert using Z-transform table term by term

```
> N2 := invztrans(z/(z-1)^2,z,n);
  N3 := factor(invztrans(z/(z-1)^3,z,n));
  N4 := factor(invztrans(z/(z-1)^4,z,n));
  N5 := factor(invztrans(z/(z-1)^5,z,n));
```

$$N2 := n$$
$$N3 := \frac{1}{2}\,n\,(n-1)$$
$$N4 := \frac{1}{6}\,n\,(n-1)\,(n-2)$$

$$N5 := \frac{1}{24} n (n-1) (n-2) (n-3) \quad (1.12)$$

```
> Rsolved := N2 + 7*N3 + 12*N4 + 6*N5 ;
```

$$Rsolved := n + \frac{7}{2} n (n-1) + 2 n (n-1) (n-2) + \frac{1}{4} n (n-1) (n-2) (n-3) \quad (1.13)$$

```
> factor(simplify(Rsolved)) ;
```

$$\frac{1}{4} n^2 (n+1)^2 \quad (1.14)$$

Example 2 (system of recurrence)

The recurrence (system of)

```
> R := x(n+1)=y(n)+1+x(n),
       y(n+1)=x(n+1)+n ;
```

$$R := x(n+1) = y(n) + 1 + x(n), y(n+1) = x(n+1) + n \quad (2.1)$$

The initial condition

```
> INI := x(0) = 1,
         y(0) = 0 ;
```

$$INI := x(0) = 1, y(0) = 0 \quad (2.2)$$

Solve using Maple primitive

```
> SOL := rsolve( {R,INI}, {x,y} ) ;
```

$$SOL := \{x(n) = 2\, 2^n - 1 - n, y(n) = 2\, 2^n - 2\} \quad (2.3)$$

Solve the same problem using Z-transform

Z-transform the recurrence

```
> EQ1 := ztrans(R[1],n,z) ;
  EQ2 := ztrans(R[2],n,z) ;
```

$$EQ1 := z\, ztrans(x(n), n, z) - x(0)\, z = ztrans(y(n), n, z) + \frac{z}{z-1} + ztrans(x(n), n, z)$$

$$EQ2 := z\, ztrans(y(n), n, z) - y(0)\, z = z\, ztrans(x(n), n, z) - x(0)\, z + \frac{z}{(z-1)^2} \quad (2.4)$$

Apply boundary conditions

```
> EQ1_BC := subs(INI,EQ1) ;
  EQ2_BC := subs(INI,EQ2) ;
```

$$EQ1\_BC := z\, ztrans(x(n), n, z) - z = ztrans(y(n), n, z) + \frac{z}{z-1} + ztrans(x(n), n, z)$$

$$EQ2\_BC := z\, ztrans(y(n), n, z) = z\, ztrans(x(n), n, z) - z + \frac{z}{(z-1)^2} \quad (2.5)$$

Solve recurrence in z with multiple step to not destroy partial factorization

```
> X_SOL := solve( EQ1_BC, {ztrans(x(n), n, z)} ) ;
```

$$X\_SOL := \left\{ ztrans(x(n), n, z) = \frac{z\, ztrans(y(n), n, z) - ztrans(y(n), n, z) + z^2}{(z-1)^2} \right\} \quad (2.6)$$

```
> Y_SOL := solve( subs(X_SOL, EQ2_BC ), {ztrans(y(n), n, z)});
```

(2.7)

$$Y\_SOL := \left\{ ztrans(y(n), n, z) = \frac{2z}{(z-1)(-2+z)} \right\} \quad (2.7)$$

```
> X_SOL := subs( Y_SOL, X_SOL ) ;
```

$$X\_SOL := \left\{ ztrans(x(n), n, z) = \frac{\frac{2z^2}{(z-1)(-2+z)} - \frac{2z}{(z-1)(-2+z)} + z^2}{(z-1)^2} \right\} \quad (2.8)$$

```
> X_SOL := simplify(X_SOL);
```

$$X\_SOL := \left\{ ztrans(x(n), n, z) = \frac{(z^2-2z+2)z}{(-2+z)(z-1)^2} \right\} \quad (2.9)$$

Perform partial fraction decomposition for X and Y

```
> X_PAR_FRAC := expand(convert(rhs(X_SOL[1])/z,parfrac)*z) ;
```

$$X\_PAR\_FRAC := \frac{2z}{-2+z} - \frac{z}{(z-1)^2} - \frac{z}{z-1} \quad (2.10)$$

```
> Y_PAR_FRAC := expand(convert(rhs(Y_SOL[1])/z,parfrac)*z) ;
```

$$Y\_PAR\_FRAC := \frac{2z}{-2+z} - \frac{2z}{z-1} \quad (2.11)$$

Invert using Z-transform table term by term

```
> T1 := invztrans(z/(z-1),z,n);
  T2 := factor(invztrans(z/(z-1)^2,z,n));
  T3 := factor(invztrans(z/(z-2),z,n));
```

$$T1 := 1$$

$$T2 := n$$

$$T3 := 2^n \quad (2.12)$$

```
> XSOL := -T1-T2+2*T3 ;
```

$$XSOL := 2\,2^n - 1 - n \quad (2.13)$$

```
> YSOL := -2*T1 + 2*T3 ;
```

$$YSOL := 2\,2^n - 2 \quad (2.14)$$

Example 3 (approximation of y''=1)

The recurrence (system of)

```
> R := h^2=y(n+2)-2*y(n+1)+y(n);
```

$$R := h^2 = y(n+2) - 2y(n+1) + y(n) \quad (3.1)$$

```
> h := 1/N ;
```

$$h := \frac{1}{N} \quad (3.2)$$

The boundary condition

```
> BC := y(0) = 0, y(N)=0 ;
```

$$BC := y(0) = 0, y(N) = 0 \quad (3.3)$$

Solve using Maple primitive (for N=10)

```
> SOL := rsolve( subs(N=10,{R,BC}), {y} ) ;
```

(3.4)

$$SOL := \left\{ y(n) = -y(9)\,(n+1) + 11\,y(9) + \frac{1}{100}\,(n+1)\left(\frac{1}{2}\,n+1\right) - \frac{11}{100}\,n + \frac{11}{25} \right\} \tag{3.4}$$

Solve the same problem using Z-transform

Z-transform the recurrence

```
> EQ := ztrans(R,n,z) ;
```

$$EQ := \frac{z}{N^2\,(z-1)} = z^2\,ztrans(y(n), n, z) - y(0)\,z^2 - y(1)\,z - 2\,z\,ztrans(y(n), n, z) + 2\,y(0)\,z + ztrans(y(n), n, z) \tag{3.5}$$

Apply boundary conditions (only initial because N is not known), y(1) will be a free parameter to set for y(N)=0

```
> EQ_BC := subs(BC,EQ) ;
```

$$EQ\_BC := \frac{z}{N^2\,(z-1)} = z^2\,ztrans(y(n), n, z) - y(1)\,z - 2\,z\,ztrans(y(n), n, z) + ztrans(y(n), n, z) \tag{3.6}$$

Solve recurrence

```
> SOL := solve( EQ_BC, {ztrans(y(n), n, z)} ) ;
```

$$SOL := \left\{ ztrans(y(n), n, z) = \frac{z\left(1 + y(1)\,N^2\,z - y(1)\,N^2\right)}{N^2\,(z-1)\left(z^2 - 2\,z + 1\right)} \right\} \tag{3.7}$$

Perform partial fraction decomposition

```
> expand(convert(rhs(SOL[1])/z,parfrac,z)*z);
```

$$\frac{z}{(z-1)^3\,N^2} + \frac{z\,y(1)}{(z-1)^2} \tag{3.8}$$

Invert using Z-transform table term by term

```
> T2 := factor(invztrans(z/(z-1)^2,z,n));
  T3 := factor(invztrans(z/(z-1)^3,z,n));
```

$$T2 := n$$

$$T3 := \frac{1}{2}\,n\,(n-1) \tag{3.9}$$

```
> YSOL := y(1)*T2 + 1/N^2*T3 ;
```

$$YSOL := y(1)\,n + \frac{1}{2}\,\frac{n\,(n-1)}{N^2} \tag{3.10}$$

Put condition for y(N)=0

```
> FINAL := subs(n=N,YSOL)=0;
```

$$FINAL := y(1)\,N + \frac{1}{2}\,\frac{N-1}{N} = 0 \tag{3.11}$$

Solve final boundary condition

```
> FINAL_SOLVED := solve( FINAL, {y(1)} ) ;
```

$$FINAL\_SOLVED := \left\{ y(1) = -\frac{1}{2}\,\frac{N-1}{N^2} \right\} \tag{3.12}$$

Put final boundary condition for the complete solution

```
> YSOL_COMPLETE := simplify(subs( FINAL_SOLVED, YSOL ));
```

$$YSOL\_COMPLETE := -\frac{1}{2}\,\frac{n\,(N-n)}{N^2} \tag{3.13}$$

```
> plot( subs(N=5,YSOL_COMPLETE), n=0..5 ) ;
```



Example 4 (approximation of y'=y-x with Heun)

```
> restart:
```

The recurrence

```
> R := y(n+1)=y(n)+(h/2)*(y(n)-x(n)+( y(n)+h*(y(n)-x(n))-x(n) )),
       x(n+1)=x(n)+h ;
```

$$R := y(n+1) = y(n) + \frac{1}{2}\,h\,(2\,y(n) - 2\,x(n) + h\,(y(n) - x(n))), x(n+1) = x(n) + h \tag{4.1}$$

The initial condition

```
> INI := x(0)=0,y(0)=0 ;
```

$$INI := x(0) = 0, y(0) = 0 \tag{4.2}$$

Solve using Maple primitive

```
> SOL := rsolve( {R,INI}, {x,y} ) ;
```

$$SOL := \left\{ x(n) = h\,n, y(n) = -\frac{-2\,h\,n - h^2\,n - 2 + 2\left(h + \frac{1}{2}\,h^2 + 1\right)^n}{2+h} \right\} \quad (4.3)$$

Solve the same problem using Z-transform

Z-transform the recurrence

```
> EQ1 := ztrans(R[1],n,z) ;
  EQ2 := ztrans(R[2],n,z) ;
```

$$EQ1 := z\,ztrans(y(n), n, z) - y(0)\,z = ztrans(y(n), n, z) + \frac{1}{2}\,h\,(2\,ztrans(y(n), n, z) - 2\,ztrans(x(n), n, z) + h\,(ztrans(y(n), n, z) - ztrans(x(n), n, z)))$$

$$EQ2 := z\,ztrans(x(n), n, z) - x(0)\,z = ztrans(x(n), n, z) + \frac{h\,z}{z-1} \quad (4.4)$$

Apply initial conditions

```
> EQ1_INI := subs(INI,EQ1) ;
  EQ2_INI := subs(INI,EQ2) ;
```

$$EQ1\_INI := z\,ztrans(y(n), n, z) = ztrans(y(n), n, z) + \frac{1}{2}\,h\,(2\,ztrans(y(n), n, z) - 2\,ztrans(x(n), n, z) + h\,(ztrans(y(n), n, z) - ztrans(x(n), n, z)))$$

$$EQ2\_INI := z\,ztrans(x(n), n, z) = ztrans(x(n), n, z) + \frac{h\,z}{z-1} \quad (4.5)$$

Solve recurrence in term of z-trasformed variables

```
> SOL := factor(solve( {EQ1_INI,EQ2_INI}, {ztrans(x(n), n, z),
  ztrans(y(n), n, z)} )) ;
```

$$SOL := \left\{ ztrans(x(n), n, z) = \frac{h\,z}{(z-1)^2}, ztrans(y(n), n, z) = \frac{h^2\,z\,(2+h)}{(z-1)^2\,(-2\,z + 2 + 2\,h + h^2)} \right\} \quad (4.6)$$

```
> YSOL := subs( SOL, ztrans(y(n), n, z) ) ;
```

$$YSOL := \frac{h^2\,z\,(2+h)}{(z-1)^2\,(-2\,z + 2 + 2\,h + h^2)} \quad (4.7)$$

```
> expand(convert( YSOL/z, parfrac, z)*z) ;
```

$$\frac{2\,z}{(2+h)\,(z-1)} + \frac{h\,z}{(z-1)^2} - \frac{4\,z}{(2+h)\,(2\,z - 2 - 2\,h - h^2)} \quad (4.8)$$

Invert using Z-transform table term by term

```
> T1 := invztrans(z/(z-1),z,n);
  T2 := factor(invztrans(z/(z-1)^2,z,n));
  T3 := factor(invztrans(z/(z-a),z,n));
```

$$T1 := 1$$

$$T2 := n$$

(4.9)

$$T3 := a^n \tag{4.9}$$

The solution is

```
> XSOL := T2*h ;
```

$$XSOL := h\,n \tag{4.10}$$

```
> YSOL := T2*h - subs(a=1+h+h^2/2,T3)*4/(2+h)/h + 2/(2+h)*T1 ;
```

$$YSOL := h\,n - \frac{4\left(h + \frac{1}{2}\,h^2 + 1\right)^n}{(2+h)\,h} + \frac{2}{2+h} \tag{4.11}$$

```
> plot( subs(h=0.01,[XSOL,YSOL,n=0..10]) ) ;
```